# CDS Table Functions and AMDP BAdIs

**Emanuel Klenner SAP Labs NSQ - May 2018**



# Terminology

- **CDS: Core Data Services - The successor of the ABAP dictionary, e.g. CDS Views, Access Controls and Table Functions**
- **VDM: Virtual Data Model for FIORI apps - Based on a 3 tiered CDS View approach using Basic Views, Transactional Views and Consumption Views**
- **BOPF: Business Object Processing Framework**
- **FIORI Elements, aka Smart Templates: Generated FIORI UIs based upon the 3 tier VDM model using the BOPF**
- **BAdI: Business Add In - Customer Exit for SAP delivered coding**
- **AMDP: ABAP Managed Database Procedures - Methods of ABAP OO classes that execute non-ABAP coding, e.g. HANA SQL Script**
- **AMDP BAdI: Customer Exit in a non-ABAP environment**
- **FNOL: First Notice of Loss - Insurance Term for the First Report of a Claim, e.g. Auto Accident**

# The Use Case – FNOL Fiori Elements App Contract Search



# Search Result List




The challenge: The insurance contract data can sit on a local claims system or on a remote policy management system and is only accessible via CDS view(s)

# ABAP Quiz ☺

```
 1 REPORT z[redacted].
 2
 3 DATA gv_polh TYPE bu_partner.
 4 DATA go_salv TYPE REF TO cl_salv_table.
 5
 6 SELECT-OPTIONS p_polh FOR gv_polh.
 7
 8 START-OF-SELECTION.
 9   BREAK-POINT.
10
11   SELECT FROM p_insurclm1stnotifcontractvh
12   FIELDS *
13   WHERE insurancepolicyholder IN @p_polh
14   INTO TABLE @DATA(gt_contract).
15
```

Is it possible that this SQL statement can produce different results when run twice on the same system with the same input parameters?

No DB changes will take place between the two executions.

# To Answer the Question …

- **We need to look at the view/table**
- **… which turns out to be a Table Function**

```
1  @ClientHandling.type: #CLIENT_DEPENDENT
2  @ClientHandling.algorithm: #SESSION_VARIABLE
3  @AccessControl.authorizationCheck: #NOT_REQUIRED
4  @VDM.viewType: #COMPOSITE
5  @VDM.private: true
6
7  define table function P_InsurClm1stNotifContractVH
8    with parameters
9      @Environment.systemField: #CLIENT
10     P_SAPClient : abap.clnt
11 returns
12 {
13   SAPclient               : abap.clnt;
14   InsuranceContract       : icl_policy;
15   InsurancePolicyholder   : bu_partner;
16   InsurancePolicy         : icl_pmi_polex;
17
18 }
19 implemented by method
20   cl_icl_fiori_fnol_amdp=>get_contract
```

"View Fields"

Data Provider

# AMDP Class Definition

CL_ICL_FIORI_FNOL_AMDP ▸ GET_CONTRACT

```
1  CLASS cl_icl_fiori_fnol_amdp DEFINITION
2    PUBLIC
3    FINAL
4    CREATE PUBLIC .
5
6    PUBLIC SECTION.
7
8      INTERFACES if_amdp_marker_hdb .
9
10     CLASS-METHODS get_contract
11         FOR TABLE FUNCTION p_insurclm1stnotifcontractvh .
12   PROTECTED SECTION.
13   PRIVATE SECTION.
14 ENDCLASS.
15
16 CLASS cl_icl_fiori_fnol_amdp IMPLEMENTATION.
17
18   METHOD get_contract
19         BY DATABASE FUNCTION
20         FOR HDB
21         LANGUAGE SQLSCRIPT
22         OPTIONS READ-ONLY
23         USING icl_fiori_fnol_amdp=>get_contract.
24
25     declare lt_result table( sapclient             "$ABAP.type( mandt )",
26                              insurancecontract     "$ABAP.type( icl_policy )",
27                              insurancepolicyholder "$ABAP.type( bu_partner )",
28                              insurancepolicy       "$ABAP.type( icl_pmi_polex )"
29                            );
30
31 *   AMDP BADI Call -> SE18 Enhancement Spot ICL_FIORI_FNOL -> BAdI Definition ICL_FIORI_FNOL_AMDP
32 *   The default implementation CL_IMP_ICL_FIORI_FNOL_AMDP selects local contracts
33 *   The secondary implemantation CL_IMP_ICL_FIORI_FNOL_AMDP_EXT shows an example how a remote contract could be retrieved
34     CALL "ICL_FIORI_FNOL_AMDP=>GET_CONTRACT"(p_sapclient=>:p_sapclient, et_contract=>:lt_result);
35
36     RETURN SELECT * FROM :lt_result;
37
38   ENDMETHOD.
39 ENDCLASS.
```

# AMDP BAdI Definition



# F1 Help for AMD BAdI Field



## AMDP BAdI Definition Restriction

Option to use a BAdI definition in an ABAP-Managed Database Procedure (AMDP).

### Use

BAdI definitions selected in this way can be used in database procedures in order to call AMDP methods of the corresponding BAdI implementation.

### Dependencies

AMDP BAdI definitions and their implementations are subject to special restrictions that are taken into consideration during the check. These include:

- Filters are not supported for AMDP BAdIs.
- Fallback classes must be specified.
- BAdI interface methods must be AMDP methods. Mixing interface methods that are written in ABAP is not allowed.
- In a BAdI class, only methods for a platform can be implemented.

# BAdI Interface

IF_EX_ICL_FIORI_FNOL_AMDP

```
1  INTERFACE if_ex_icl_fiori_fnol_amdp
2    PUBLIC .
3
4
5    INTERFACES if_badi_interface .
6
7    CLASS-METHODS get_contract
8      IMPORTING
9        VALUE(p_sapclient) TYPE mandt
10     EXPORTING
11       VALUE(et_contract) TYPE icl_fiori_fnol_contract_t .
12
13 ENDINTERFACE.
```

# BAdI Default Implementation – Accessing Local System

CL_IMP_ICL_FIORI_FNOL_AMDP ▸ IF_EX_ICL_FIORI_FNOL_AMDP~GET_CONTRACT

```
CLASS cl_imp_icl_fiori_fnol_amdp DEFINITION
  PUBLIC
  CREATE PUBLIC .
  PUBLIC SECTION.
    INTERFACES if_badi_interface .
    INTERFACES if_ex_icl_fiori_fnol_amdp .
    INTERFACES if_amdp_marker_hdb.

  PROTECTED SECTION.
  PRIVATE SECTION.
ENDCLASS.

CLASS cl_imp_icl_fiori_fnol_amdp IMPLEMENTATION.

  METHOD if_ex_icl_fiori_fnol_amdp~get_contract
         BY DATABASE PROCEDURE
         FOR HDB
         LANGUAGE SQLSCRIPT
         OPTIONS READ-ONLY
         USING iclclaim
               iclpartocc.

    et_contract =
      select
        a.client   as sapclient,
        a.policy   as insurancecontract,
        b.bpartner as insurancepolicyholder,
              ' '  as insurancepolicy
      from
        iclclaim        as a
        join iclpartocc as b
        on   b.client = a.client
        and  b.claim  = a.claim
      where  a.client       = :p_sapclient
      and    a.claim        =  a.policy
      and    b.subobjcat    = 'PO'
      and    b.subobject    = ' '
      and    b.role         = 'POLH'
      and    b.deleted     <> 'X'
      order by a.policy;

  ENDMETHOD.
ENDCLASS.
```

# BAdI Implementation II – Accessing a Remote System

ZCL_IMP_ICL_FIORI_FNOL_AMDP_EX

```
1  CLASS zcl_imp_icl_fiori_fnol_amdp_ex DEFINITION
2  PUBLIC
3    CREATE PUBLIC .
4
5    PUBLIC SECTION.
6
7      INTERFACES if_badi_interface .
8      INTERFACES if_ex_icl_fiori_fnol_amdp .
9      INTERFACES if_amdp_marker_hdb .
10   PROTECTED SECTION.
11   PRIVATE SECTION.
12 ENDCLASS.
13
14
15
16 CLASS zcl_imp_icl_fiori_fnol_amdp_ex IMPLEMENTATION.
17
18   METHOD if_ex_icl_fiori_fnol_amdp~get_contract
19           BY DATABASE PROCEDURE
20           FOR HDB
21           LANGUAGE SQLSCRIPT
22           OPTIONS READ-ONLY.
23
24 *    This is what your SQL statement should look like
25     et_contract =
26       select
27           "SAPClient"               as SAPClient,
28           "APPLNR_TT"               as InsuranceContract,
29           "InsurancePolicyholder"   as InsurancePolicyholder,
30           "InsurancePolicy"         as InsurancePolicy
31       FROM
32         "_SYS_BIC"."z_sap.z_fspm/Z_POLICYHOLDER_CONTRACT"
33       WHERE
34        "SAPClient" = :P_SAPclient
35 *       AND
36 *        "ZPRCLU" IN ( 'ZV', 'MV' )
37
38     ;
39
40   ENDMETHOD.
```

# Implementation in the App – CDS Value Help View

```
define view I_InsurClm1stNotifContrValHelp
  as select from I_BusinessPartner as _PolicyHolder

  inner join P_InsurClm1stNotifContractVH(P_SAPClient:$session.client) as _Policy_PM on _PolicyHolder.BusinessPartner = _Policy_PM.InsurancePolicyholder

{
      @UI.selectionField: [ { position: 130 } ]
      @Search.defaultSearchElement : true
      @Search.fuzzinessThreshold : 0.8
      @Search.ranking : #HIGH
  key InsuranceContract,

      @UI.selectionField: [ { position: 10 } ]
      @Search.defaultSearchElement : true
      @Search.fuzzinessThreshold : 0.8
      @Search.ranking : #HIGH
      _PolicyHolder.BirthDate,

      @UI.selectionField: [ { position: 30 } ]
      @Search.defaultSearchElement : true
      @Search.fuzzinessThreshold : 0.8
      @Search.ranking : #HIGH
      @Semantics.name.familyName: true
      _PolicyHolder.LastName,

      @UI.selectionField: [ { position: 20 } ]
      @Search.defaultSearchElement : true
      @Search.fuzzinessThreshold : 0.8
      @Search.ranking : #MEDIUM
      @Semantics.name.givenName: true
      _PolicyHolder.FirstName,

      @UI.selectionField: [ { position: 40 } ]
      @Semantics.name.fullName: true
      cast( _PolicyHolder.BusinessPartnerName as icl_policyholder_name preserving type)                    as InsurancePolicyholderName,

      @UI.selectionField: [ { position: 50 } ]
      @Semantics.address.type: [ { HOME } ]
      @Semantics.address.street: true
      _PolicyHolder._CurrentDefaultAddress._Address.StreetName,
```

# Value Help View in the CDS Consumption View

```
define view C_InsurClm1stNotifTP
  as select from I_InsurClm1stNotifTP

  association [1..1] to I_InsurClmIncdntType          as _IncidentType     on  _IncidentType.InsurClmPolicyProduct        = $projection.InsurClmP
                                                                           and _IncidentType.InsurClmPolicyProdGeneration = $projection.InsurClmP
                                                                           and _IncidentType.InsurClmPolicyProductVersion = $projection.InsurClmP
                                                                           and _IncidentType.InsurClmIncdntType           = $projection.InsurClmI

  association [0..*] to I_InsurClmIncdntTypeText      as _IncidentTypeText on  $projection.InsurClmIncdntType = _IncidentTypeText.InsurClmIncdntT

  association [0..1] to I_Country                     as _Country          on  _Country.Country = $projection.InsurClmLossLocCountry

  association [0..1] to I_Region                      as _Region           on  _Region.Region  = $projection.InsurClmLossLocRegion
                                                                           and _Region.Country = $projection.InsurClmLossLocCountry

  association [0..*] to C_InsurClm1stNotifNoteTP      as _Note             on  $projection.InsuranceClaim = _Note.InsuranceClaim

  association [0..*] to C_InsurClm1stNotifDmgdVehTP   as _DamagedVehicle   on  $projection.InsuranceClaim = _DamagedVehicle.InsuranceClaim

  association [0..*] to C_InsurClm1stNotifWtnssTP     as _Witness          on  I_InsurClm1stNotifTP.InsuranceClaim = _Witness.InsuranceClaim

  association [0..1] to I_InsurClmHandler             as _ClaimHandler     on  $projection.InsuranceClaim = _ClaimHandler.InsuranceClaim

  association [1..1] to I_InsurClm1stNotifIncdntTypeVH as _InciVH          on  $projection.InsurClmIncdntType = _InciVH.InsurClmIncdntType

  association [0..1] to I_InsurClm1stNotifCauseVH     as _CauseVH          on  $projection.InsurClmCauseOfLoss = _CauseVH.InsurClmCauseOfLoss

  association [1..1] to I_InsurClm1stNotifBP          as _PolicyHolder     on  _PolicyHolder.BusinessPartner = $projection.InsurancePolicyholder

  association [1..1] to I_InsurClm1stNotifContrValHelp as _ContractVH      on  _ContractVH.InsuranceContract = $projection.InsuranceContract

  association [0..*] to C_InsurClm1stNotifFactsTP     as _Facts          on  $projection.InsuranceClaim = _Facts.InsuranceClaim

{

      // ============== Start of Claim General Information ====================
      @ObjectModel.readOnly: true
  key InsuranceClaim,
      //      @Consumption.valueHelp: '_Contract'
      @Consumption.valueHelp: '_ContractVH'
      InsuranceContract,
```

# Sequence Diagram I

Start FNOL app

Create New Claim(Draft)

Start typing Policy Holder Name in Contract Field

CDS view **I_InsurClm1stNotifContrValHelp** searches for BP

View joins to Table Function **P_InsurClm1stNotifContractVH**

Table Function calls AMDP Method
**CL_ICL_FIORI_FNOL_AMDP=>GET_CONTRACT**

# Sequence Diagram II

Default Implementation of BAdI **ICL_FIORI_FNOL_AMDP** active

Secondary implementation of BAdI active

**CL_IMP_ICL_FIORI_FNOL_AMDP**

**CL_IMP_ICL_FIORI_FNOL_AMDP_EXT**

Method **GET_CONTRACT**

Method **GET_CONTRACT**

Selects local contract via Join ICLCLAIM to ICLPARTOCC

Selects contract via remote access to view pointing to FS-PM

User selects contract number from drop down list box and hits the ENTER key

# Thank You!